Controlling Office 2003 Hyperlinks

One of the most interesting features introduced back in Office 97 was the capability to create hyperlinks in any kind of Office document: Word documents, Excel worksheets, Access databases, PowerPoint presentations, and Outlook email messages. This lets you create "active" documents that let the reader click special text sections and "surf" to another document, which might be on the World Wide Web, your corporate intranet, or your hard drive.

For example, consider the Word text shown below:

Last National Bank

1234 Mammon Way

Croesus, CA  90219

Wednesday, October 22, 2003
Harold G. Longsuffering

55 Hard Luck Road

Downanout, CA, 90210

Dear Mr. Longsuffering:
I am happy to report that your loan has been approved! (That little “present” you sent to me sure made a big difference, wink, wink.) Based on the principle amount of $10,000 loaned over 48 months at 8% interest per annum, your monthly payment is $244.13. The first payment is due on the 1st of next month.

Note that I have posted an amortization schedule on our Web site should you wish to view the particulars of the loan.

Please feel free to contact me if you have any questions. Thank you for choosing Last National Bank.

Sincerely,
Sly M. Baul

Note that the phrase "amortization schedule" is displayed underlined and in a different color (blue). This formatting indicates that this phrase is a hyperlink.

Tip: You can see where a hyperlink will take you by moving the mouse pointer over the link text. After a second or two, a banner appears that tells you the name of the hyperlink document.

The Office products boast a number of different methods for creating these hyperlinks, but the most common is to enter the appropriate information by hand by selecting the Insert, Hyperlink command. The application displays the Insert Hyperlink dialog box, so that you can define the particulars of the hyperlink, including the network path or Web address, and a named location (such as a bookmark) within the file.

Working with the Hyperlink Object

Each Office application's object model includes the Hyperlink object, which represents a hyperlink in a document. Using VBA, you can insert Hyperlink objects into a document, add hyperlinks to the Favorites folder, and even "surf" the hyperlinks.

Hyperlinks is the object that represents the collection of all hyperlinks in an entire document or in part of a document (such as a range). You use Hyperlinks(Index) to refer to a specific hyperlink, where Index can be either of the following:

· A number that represents the Hyperlink object, where 1 represents the first hyperlink, 2 the second hyperlink, and so on.

· A string that contains the hyperlink address (a file path, UNC path, or a URL).

Adding New Hyperlink Objects

The Hyperlinks collection has an Add method that you use to insert a new Hyperlink object into a document:

Object.Hyperlinks.Add(Anchor, Address, SubAddress, ScreenTip, TextToDisplay, Target)

Object
The object in which you want to insert the hyperlink (such as a Document or Range object in Word, or a Worksheet or Range object in Excel).

Anchor
The object the user will click to follow the hyperlink. This could be a section of text, a Range object, or a graphic.

Address
The address of the link, which can be a file path, a UNC path, or a URL.

SubAddress
(optional) A named location within the linked file (such as a bookmark in a Word document or a named range in an Excel workbook).

ScreenTip
(optional) A string that specifies the banner text that appears when the user hovers the mouse pointer over the link. If you omit this argument, the Address is displayed.

TextToDisplay
(optional) A string that specifies the hyperlink text that appears in the document. This will replace whatever is specified by Anchor.

Target
(optional) A string that specifies the name of a frame or of an open window in which you want the linked document to appear.

Listing 1 shows a procedure that uses the Add method to insert a link to a URL in a Word document.

Listing 1. A Procedure that Adds a Hyperlink Object

Sub AddLink()

    Dim r As Range

    '

    ' Add a paragraph to the end of the document

    '

    With ThisDocument.Paragraphs

        .Item(.Count).Range.InsertParagraphAfter

        Set r = .Item(.Count).Range

    End With

    r.Text = "Sams' Home Page"

    r.Hyperlinks.Add _

        Anchor:=r, _

        Address:="http://www.samspublishing.com/", _

        ScreenTip:="Click to visit the home page of Sams!"

End Sub

Hyperlink Object Properties

Once you've added a hyperlink or two, you can use the properties of the Hyperlink object to manipulate these links as needed. Here's a quick rundown of some useful properties:

Hyperlink.Address: Returns or sets the address associated with the specified Hyperlink.

Hyperlink.EmailSubject: Returns or sets the subject line if the specified Hyperlink sends an e-mail message (that is, the address begins with "mailto" instead of "http").

Hyperlink.Range: Returns the Range object associated with the specified Hyperlink.

Hyperlink.ScreenTip: Returns or sets the screen tip associated with the specified Hyperlink.

Hyperlink.Shape: Returns the Shape object associated with the specified Hyperlink.

Hyperlink.SubAddress: Returns or sets the named location within the target associated with the specified Hyperlink.

Hyperlink.Target: Returns or sets the name of the frame or open window in which to display the specified Hyperlink.

Hyperlink.TextToDisplay: Returns or sets the text displayed for the specified Hyperlink.

Hyperlink.Type: Returns a number that specifies the kind of link associated with the specified Hyperlink. This property returns one of the following constants: msoHyperlinkInlineShape, msoHyperlinkRange, or msoHyperlinkShape.

Hyperlink Object Methods

The Hyperlink object has only four methods:

Hyperlink.AddToFavorites: Adds the specified Hyperlink to the current user's Favorites folder.

Hyperlink.CreateNewDocument: Creates a new document (and returns the appropriate document object) that is linked to the specified Hyperlink. Use this method when you create a hyperlink that points to a file that doesn't exist yet. Here's the syntax to use:

Hyperlinks.CreateNewDocument(Filename, EditNow, Overwrite)

Hyperlink
The Hyperlink object.

Filename
The path and filename of the new document.

EditNow
A Boolean value. Use True to open the new document for editing; use False to not open the document.

Overwrite
A Boolean value. Use True to have the new document replace an existing document with the same name, if one exists; use False to prevent an existing document from being replaced.

Hyperlink.Delete: Deletes the specified Hyperlink.

Hyperlink.Follow: Displays the linked document associated with the specified Hyperlink. Here's the syntax to use:

Hyperlinks.Follow(NewWindow, AddHistory, ExtraInfo, Method, HeaderInfo)

Hyperlink
The Hyperlink object.

NewWindow
(optional) Set this argument to True to display the document in a new window. Use False (the default) to display the document in the current window. (Note that this depends on the application used to display the linked document.)

AddHistory
(optional) This argument is reserved for future use.

ExtraInfo
(optional) Use this argument to specify additional information that HTTP requires to resolve the hyperlink. The most common uses for this argument are to send a Web server the contents of a form, the coordinates of an image map, or a search parameter for an ASP file.

Method
(optional) This argument determines how the data specified with ExtraInfo is sent to the server. Use msoMethodGet to append the data to the link address; use mspMethodPost to post the data either as a String or as a Byte array (depending on the data type of the data).

HeaderInfo
(optional) Use this argument to specify header data for the HTTP header.

For example, the Google Web site is a search engine that lets you query its! database of Web pages. Here's the general syntax for a Google search query:

http://www.google.com/bin/search?q=keyword
Here, keyword is a keyword you want to use as the criterion for your search. If you wanted to implement this in the Follow method, the part after the question mark (?) would correspond to the ExtraInfo argument, and because this data is appended to the address, the Method argument must be msoMethodGet.

To demonstrate this, Listing 2 contains two procedures: AddGoogle creates a link to the Google search engine and SearchGoogle prompts for a keyword and then runs the Follow method.

Listing 2. Procedures that Add a Link for the Google Search Engine and Run a Query on the Google Database

Sub AddGoogle()

    Dim r As Range

    '

    ' Add a paragraph to the end of the document

    '

    With ThisDocument.Paragraphs

        .Item(.Count).Range.InsertParagraphAfter

        Set r = .Item(.Count).Range

    End With

    r.Text = "Google Search"

    r.Hyperlinks.Add _

        Anchor:=r, _

        Address:="http://www.google.com/search"

End Sub

Sub SearchGoogle()

    Dim link As Hyperlink

    Dim keyword As String

    Set link = ThisDocument.Hyperlinks("http://www.google.com/search")

    keyword = InputBox("Enter a search keyword:")

    link.Follow _

        extrainfo:="q=" & keyword, _

        Method:=msoMethodGet

End Sub

The FollowHyperlink Method

You saw in the preceding section that you can use the Follow argument to display a target document associated with an existing Hyperlink object. What do you do, however, if you would rather not clutter a document with hyperlinks? For these situations, the Office applications each have a FollowHyperlink method that displays a target document without requiring an existing Hyperlink object. Here are the various syntaxes:

Word:

Document.FollowHyperlink(Address, SubAddress, NewWindow, AddHistory, ExtraInfo, Method, HeaderInfo)

Excel:

Workbook.FollowHyperlink(Address, SubAddress, NewWindow, AddHistory, ExtraInfo, Method, HeaderInfo)

PowerPoint:

Presentation.FollowHyperlink(Address, SubAddress, NewWindow, AddHistory, ExtraInfo, Method, HeaderInfo)

Access:

Application.FollowHyperlink(Address, SubAddress, NewWindow, AddHistory, ExtraInfo, Method, HeaderInfo)

For example, Listing 3 shows how you'd use this method in Word to perform the same Google search you saw earlier.

Listing 3. Using the FollowHyperlink Method to Display a Target Document Without an Existing Hyperlink Object

Sub FollowHyperlinkTest()

    Dim keyword As String

    keyword = InputBox("Enter a search keyword:")

    ThisDocument.FollowHyperlink _

        Address:="http://www.google.com/search", _

        ExtraInfo:="q=" & keyword, _

        Method:=msoMethodGet

End Sub

